

Deriving proved equality tests in Coq-elpi

Stronger induction principles for containers in Coq

Enrico Tassi

Université côte d’Azur - Inria

Enrico.Tassi@inria.fr

Abstract

We describe a procedure to derive equality tests and their correctness proofs from inductive type declarations. Programs and proofs are derived compositionally, reusing code and proofs derived previously.

The key steps are two. First, we design appropriate induction principles for data types defined using parametric containers. Second, we develop a technique to work around the modularity limitations imposed by the purely syntactic termination check Coq performs on recursive proofs. The unary parametricity translation of inductive data types turns out to be the key to both steps.

Last but not least, we provide an implementation of the procedure for the Coq proof assistant based on the Elpi [3] extension language.

Keywords Coq, Containers, Induction, Equality test, Parametricity translation

1 Introduction

Modern typed programming languages come with the ability of generating boilerplate code automatically. Typically when a data type is declared a substantial amount of code is made available to the programmer at little cost, code such as an equality test, a printing function, generic visitors etc. For example the `derive` directive of Haskell or the `ppx_deriving` OCaml preprocessor provide these features for the respective programming language.

The situation is less than ideal in the Coq proof assistant. It is capable of synthesizing the recursor of a datatype, that, following the Curry-Howard isomorphism, implements the induction principle associated to that datatype. It supports all datatypes, containers such as lists included, but generates a quite disappointing principle when a datatype *uses* a container.

For example, let’s take the data type rose tree, where \mathcal{U} stands for a universe (such as `Prop` or `Type`):

```
Inductive rtree A :  $\mathcal{U}$  :=  
  | Leaf (a : A)  
  | Node (l : list (rtree A)).
```

Its associated induction principle is the following one:

```
1 Lemma rtree_ind :  $\forall A$  ( $P$  :  $\text{rtree } A \rightarrow \mathcal{U}$ ),  
2   ( $\forall a$  :  $A$ ,  $P$  (Leaf  $A$   $a$ ))  $\rightarrow$   
3   ( $\forall l$  : list ( $\text{rtree } A$ ),  $P$  (Node  $A$   $l$ ))  $\rightarrow$   
4    $\forall r$  :  $\text{rtree } A$ ,  $P$   $r$ .
```

Remark that the recursive step, line 3, lacks any induction hypotheses on (the elements of) l while one would expect P to hold on each and every subtree. Even a very basic recursive program such as an equality test cannot be proved correct using this induction principle. To be honest, the Coq user is not even supposed to write equality tests by hand, nor to prove them correct interactively. Coq provides two facilities to synthesize equality tests and their correctness proofs called `Scheme Equality` and `decide equality`. The former is fully automatic but is unfortunately very limited, for example it does not support containers. The latter requires human intervention and generates a single, very large, term that mixes code and proofs.

As a consequence, users often need to manually write induction principles, equality tests and their correctness proofs. This situation is very unfortunate because the need for the automatic generation of boilerplate code such as equality tests is higher than ever in the Coq ecosystem. All modern formal libraries structure their contents in a hierarchy of interfaces and some machinery such as type classes [14] or canonical structures [5] are used to link the abstract library to the concrete instances the user is working on. For example the first interface one is required to implement in order to use the theorems in Mathematical Components library [6] on a type τ is the `eqType` one, that requires a correct equality test on τ .

In this paper we use the framework for meta programming based on Elpi [3, 15] developed by the author and we focus on the derivation of equality tests. It turns out that generating equality tests is relatively easy, while their correctness proofs are hard to synthesise, for two reasons. The first problem is that the standard induction principles generated by Coq, as shown before, are too weak. In order to strengthen them one needs quite some extra boilerplate, such as the derivation of the unary parametricity translation of the data types involved. The second reason is that termination checking is purely syntactic in Coq. Rephrased along the Curry-Howard isomorphism this means that in order to check that the induction hypothesis is applied to a smaller term, Coq may need to unfold all theorems involved

in the proof. This, in practice, forces all proofs to be transparent breaking modularity: a statement is no more a contract, changing its proof script may impact users.

In this paper we describe a derivation procedure for equality tests and their correctness proofs where programs and proofs are both derived compositionally, reusing code and proofs derived previously. This procedure also confines the termination check issue, allowing proofs to be mostly opaque. More precisely the contributions of this paper are the following ones:

- A technique to confine the termination checking issue out of the main proofs. In this paper we apply it to the correctness proof of equality tests, but the technique is applicable to all proofs that proceed by structural induction.
- A modular and structured process to derive proved equality tests and, en passant, stronger induction principles for inductive types defined using containers.
- An actual implementation based on the Elpi extension language for the Coq proof assistant.

Straight to the point, by installing the `coq-elpi` package¹ one obtains the following definition, where $(\text{reflect } P \ b)$ is a predicate stating the equivalence between a proposition P and a boolean test b .

```
Definition eq_axiom T f x :=
  ∀ y, reflect (x = y) (f x y).
```

Then by issuing the command `Elpi derive rtree` one gets the following terms automatically synthesized out of the type declaration for `rtree`:

```
Definition rtree_eq :
  ∀ A, (A → A → bool) → rtree A → rtree A → bool.
```

```
Lemma rtree_eq_OK : ∀ A (A_eq : A → A → bool),
  (∀ a, eq_axiom A A_eq a) →
  ∀ t, eq_axiom (rtree A) (rtree_eq A A_eq) t.
```

The former is a (transparent) equality test for `rtree`. The latter is a (opaque) proof of correctness for `rtree_eq` under the assumption that the equality test `A_eq` is correct.

The paper introduces the problem in section 2 by describing the shape of an equality test and of its correctness proof and explaining the modularity problem that stems for the termination checker of Coq. It then presents the main idea behind the modular derivation procedure in section 3. Section 4 briefly introduces the Elpi extension language and section 5 describes all the bricks composing the derivation.

2 The problem: equality test proofs meet syntactic termination checking

Recursors, or induction principles, are not primitive notions in Coq. The language provides constructors for fix point and

pattern matching that work on any inductive data the user can declare.

For example to test two lists `l1` and `l2` for equality one first takes in input an equality test `A_eq` for the elements of type `A` and then performs the recursion:

```
1 Definition list_eq A (A_eq : A → A → bool) :=
2   fix rec (l1 l2 : list A) {struct l1} : bool :=
3     match l1, l2 with
4     | nil, nil => true
5     | x :: xs, y :: ys => A_eq x y && rec xs ys
6     | _, _ => false
7   end.
```

Coq accepts this definition because the recursive call is on `xs` that is a syntactically smaller term, i.e. a subterm, of the input term `l1` (the argument labelled as decreasing by the `{struct l1}` annotation).

Let's now define the equality test for the `rtree` data type by reusing the equality test for lists:

```
8 Definition rtree_eq B (B_eq : B → B → bool) :=
9   fix rec (t1 t2 : rtree B) {struct t1} : bool :=
10     match t1, t2 with
11     | Leaf x, Leaf y => B_eq x y
12     | Node l1, Node l2 =>
13       list_eq (rtree B) rec l1 l2
14     | _, _ => false
15   end.
```

Note that `list_eq` is called passing as the `A_eq` argument the fixpoint `rec` itself (line 13). In order to check that the latter definition is sound, Coq looks at the body of `list_eq` to see whether its parameter `A_eq` is applied to a term smaller than `t1`. Since `l1` is a subterm of `t1` and since `x` is a subterm of `l1`, then the recursive call $(\text{rec } x \ y)$ at line 5 is legit.

This is pretty reasonable for programs. We want both `list_eq` and `rtree_eq` to compute, hence their body matters to us. The fact that checking the termination of `rtree_eq` requires inspecting the body of `list_eq` is not very annoying this time.

On the contrary proof terms are typically hidden to the type checker once they have been validated, for both performance and modularity reasons. The desire is to make only the statement of theorems binding, and keep the freedom to clean, refactor, simplify proofs without breaking the rest of the formal development.

For example, let's assume we proved that `list_eq` is correct.

```
1 Lemma list_eq_OK : ∀ A (A_eq : A → A → bool),
2   (∀ a, eq_axiom A A_eq a) →
3   ∀ l, eq_axiom A (list_eq A A_eq) l.
4 Proof. .. Qed.
```

It seems desirable to use this lemma in order to prove the correctness of `rtree_eq`, since it calls `list_eq`. Unfortunately the following proof is rejected if the body of `list_eq_OK` is hidden to the type checker:

¹See <https://github.com/LPCIC/coq-elpi> for the installation instructions

```

221 5 Lemma rtree_eq_OK B B_eq (HB: ∀ b, eq_axiom B B_eq b) :
222 6   ∀ t, eq_axiom (rtree B) (rtree_eq B B_eq) t
223 7 :=
224 8   fix IH (t1 t2 : rtree B) {struct t1} :=
225 9   match t1, t2 with
226 10  | Node l1, Node l2 =>
227 11   .. list_eq_OK (rtree B) (tree_eq B B_eq) IH l1 l2 ..
228 12  | Leaf b1, Leaf b2 => .. HB b1 b2 ..
229 13  | .. => ..
230 14   end.

```

We pass `IH`, the induction hypothesis, as the witness that `(tree_eq B B_eq)` is a correct equality test (the argument at line 11 preceding `IH`). Without knowing how this argument is used by `list_eq_OK`, Coq rejects the term.

The issue seems unfixable without changing Coq in order to use a more modular check for termination, for example based on sized types [12]. We propose a less ambitious but more practical approach here, that consists in putting the transparent terms that the termination checker is going to inspect outside of the main proof bodies so that they can be kept opaque.

The intuition is to “reify” the property the termination checker wants to enforce. It can be phrased as “`x` is a subterm of `t` and has the same type”. More in general we model “`x` is a subterm of `t` with property `P`”. Property “`P`” is going to be “being of the same type” for subterms of “`t`” that are of the same type, while “`P`” will be an arbitrary property for terms of an arbitrary type such as the elements of a list.

This relation is naturally expressed by the unary parametricity translation of types [17].

3 The idea: separating terms and types via the unary parametricity translation

Given an inductive type `T` we systematically name `is_T` an inductive predicate describing the type of the inhabitants of `T`. This is the one for natural numbers:

```

258 Inductive is_nat : nat → U :=
259 | is_0 : is_nat 0
260 | is_S n (pn : is_nat n) : is_nat (S n).

```

The one for a container such as `list` is more interesting:

```

262 Inductive is_list A (PA : A → U) : list A → U :=
263 | is_nil : is_list A PA nil
264 | is_cons a (pa : PA a) l (pl : is_list A PA l) :
265   is_list A PA (a :: l).

```

Remark that all the elements of the list validate `PA`.

When a type `T` is defined in terms of another other type `C`, typically a container, the `is_C` predicate shows up inside `is_T`. For example:

```

270 1 Inductive is_rtree A (PA : A → U) : rtree A → U :=
271 2 | is_Leaf a (pa : PA a) : is_rtree A PA (Leaf A a)
272 3 | is_Node l (pl : is_list (rtree A) (is_rtree A PA l)) :
273 4   is_rtree A PA (Node A l).

```

Note how line 3 expresses the fact that all elements in the list `l` validate `(is_rtree A PA)`, i.e. they are rose trees.

Our intuition is that these predicates reify the notion of being of a certain type, structurally. What we typically write `(t : T)` can now be also phrased as `(is_T t)` as one would do in a framework other than type theory, such as a mono-sorted logic.

It turns out that the inductive predicate `is_T` corresponds to the unary parametricity translation of the type `T`. Keller and Lasson in [4] give us an algorithm to synthesize these predicates automatically.

What we look for now is a way to synthesize a reasoning principle for a term `t` when `(is_T t)` holds.

3.1 Stronger induction principles for containers

Let’s have a look at the standard induction principle of lists.

```

293 Lemma list_ind A (P : list A → U) :
294   P nil →
295   (∀ a l, P l → P (a :: l)) →
296   ∀ l : list A, P l.

```

This reasoning principle is purely parametric on `A`, no knowledge on any term of type `A` such as `a` is ever available.

What we want to obtain is a more powerful principle that lets us choose some invariant for the subterms of type `A`. The one we synthesise is the following one, where the differences are underlined.

```

303 1 Lemma list_induction A (PA: A → U) (P: list A → U):
304 2   P nil →
305 3   (∀ a (pa : PA a) l, P l → P (a :: l)) →
306 4   ∀ l, is_list A PA l → P l.

```

Note the extra premise `(is_list A PA l)`: The implementation of this induction principle goes by recursion on the term of this type and finds as an argument of the `is_cons` constructor the proof evidence `(pa : PA a)` it feeds to the second premise (line 3). Intuitively all terms of type `(list A)` validate the property `P`, while all terms of type `A` validate the property `PA`.

More in general to each type we attach a property. For parameters we let the user choose (we take another parameter, `PA` here). For the type being analysed, `list A` here, we take the usual induction predicate `P`. For terms of other types we use their unary parametricity translation.

Take for example the induction principle for `rtree`.

```

320 1 Lemma rtree_induction A PA (P : rtree A → U) :
321 2   (∀ a, PA a → P (Leaf A a)) →
322 3   (∀ l, is_list (rtree A) P l → P (Node A l)) →
323 4   ∀ t, is_rtree A PA t → P t.

```

Line 3 uses `is_list` to attach a property to `l`, and given that `l` has type `(list (rtree A))` the property for the type parameter `(rtree A)` is exactly `P`. Note that this induction principle gives us access to `P`, the property one is proving, on the subtrees contained in `l`.

3.1.1 Synthesizing stronger induction principles

We postpone a detailed description of the synthesis to section 5.4, here we just sketch how to build the type on the induction principle.

It turns out that the types of the constructors of `is_T` give us a very good hint on the type of the induction principle.

The type of the first premise

$$(\forall a, PA\ a \rightarrow P\ (Leaf\ A\ a)) \rightarrow$$

is exactly the type of the `is_Leaf` constructor

$$| is_Leaf\ a\ (pa : PA\ a) : is_rtree\ A\ PA\ (Leaf\ A\ n)$$

where $(is_rtree\ A\ PA)$ is replaced by P . The same holds for the other premise: its type can be trivially obtained from the type of `is_Node`.

Our intuition is that the inductive predicate `is_T` provides the same information that typing provides. Induction principles give P on (smaller) terms of the same type, that would be terms for which `is_T` holds. Given their inductive nature, `is_T` predicates are able to propagate the desired property inside parametric containers.

3.2 Isolating the syntactic termination check

As one expects, it is possible to prove that `is_T` holds for terms of type T .

```
Definition nat_is_nat : ∀ n : nat, is_nat n :=
  fix rec n : is_nat n :=
    match n as i return (is_nat i) with
    | 0 => is_0
    | S p => is_S p (rec p)
  end.
```

For containers we can prove this class of theorems when the property on the parameter is true on the entire type.

```
Definition list_is_list : ∀ A (PA : A → U),
  (∀ a, PA a) → ∀ l, is_list A PA l.
```

```
Definition rtree_is_rtree : ∀ A (PA : A → U),
  (∀ a, PA a) → ∀ t, is_rtree A PA t.
```

These facts are then to be used in order to satisfy the premise of our induction principles.

Going back to our goal, we can build correctness proofs of equality tests in two steps. For example, for natural numbers we can generate two lemmas:

```
1 Lemma nat_eq_correct :
2   ∀ n, is_nat n → eq_axiom nat nat_eq n :=
3   nat_induction (eq_axiom nat nat_eq) P0 PS.
4
5 Lemma nat_eq_OK n : eq_axiom nat nat_eq n :=
6   nat_eq_correct n (nat_is_nat n).
```

where $P0$ and PS (line 3) stand for the two proof terms corresponding to the base case and the inductive step of the proof. We omit them because they play no role in the current discussion.

For containers we can link the pieces in a similar way. For example the correctness proof for the equality test on the $(list\ A)$ data type can be proved as follows, where again line 7 omits the proof steps for `nil` and `cons`.

```
1 Lemma list_eq_correct A A_eq :
2   ∀ l, is_list A (eq_axiom A A_eq) l →
3     eq_axiom list A (list_eq A A_eq) l
4 :=
5   list_induction A (eq_axiom A A_eq)
6   (eq_axiom (list A) (list_eq A A_eq))
7   Pnil Pcons.
8
9 Lemma list_eq_OK A A_eq (HA : ∀ a, eq_axiom A A_eq a) l :
10  eq_axiom (list A) (list_eq A A_eq) l :=
11  list_eq_correct A A_eq
12  l (list_is_list (eq_axiom A A_eq) HA l).
```

What is more interesting is to look at the correctness proof of the equality test for `rtree`. Note how the induction hypothesis $P1$ given by `rtree_induction` perfectly fits the premise of `list_eq_correct`.

```
1 Lemma rtree_eq_correct A A_eq :
2   ∀ t, is_rtree A (eq_axiom A A_eq) t →
3     eq_axiom (rtree A) (rtree_eq A A_eq)
4 :=
5   rtree_induction A (eq_axiom A A_eq)
6   (eq_axiom (rtree A) (rtree_eq A A_eq))
7   PLeaf
8   (λ l P1 : is_list (rtree A)
9     (eq_axiom (rtree A) (rtree_eq A A_eq)) l =>
10    .. list_eq_correct (rtree A) (rtree_eq A A_eq) l P1 ..)
11
12 Lemma rtree_eq_OK A A_eq (HA : ∀ a, eq_axiom A A_eq a) t :
13  eq_axiom (rtree A) (rtree_eq A A_eq) t :=
14  rtree_eq_correct A A_eq
15  t (rtree_is_rtree A (eq_axiom A A_eq) HA t).
```

Type checking the terms above does not require any term to be transparent. Actually they are applicative terms, there is no apparently recursive function involved.

Still there is no magic, we just swept the problem under the rug. In order to type check the proof of `rtree_is_rtree` Coq needs to look at the proof term of `list_is_list`:

```
1 Definition rtree_is_rtree A PA (HPA : ∀ a, PA a) :=
2   fix IH t {struct t} : is_rtree A PA t :=
3   match t with
4   | Leaf a => is_Leaf A PA a (HPA a)
5   | Node l =>
6     is_Node A PA l
7     (list_is_list (rtree A) (is_rtree A) IH l)
8   end.
```

As we explained in section 2 Coq needs to know the body of `list_is_list` in order to agree that the argument IH is only used on subterms of t .

Even if we can't make the problem disappear (without changing the way Coq checks termination), we claim we

confined the termination checking issue to the world of reified type information. The transparent proofs of theorems such as `T_is_T` are separate from the other, more relevant, proofs that can hence remain opaque as desired.

4 Elpi: an extension language for Coq

Elpi [3] is a dialect of λ Prolog [9], an higher order logic programming language. Elpi can be used as an extension language for Coq [15] in order to develop new commands in a programming language that has native support for bound variables.

Coq terms are represented in λ -tree syntax style [8] (sometimes also called Higher Order Abstract Syntax) reusing the binders of the programming language to represent the ones of Coq. For example, the term $(\lambda x \Rightarrow \text{fact } x)$ is represented as `(lam (λx , app["fact", x]))`. We say that `app` and `lam` are object level term constructors standing for iterated (n-ary) application and unary lambda abstraction; `"fact"` is a constant and `x` is a variable bound by λx , that is the binder of the programming language.²

Programs are organized in clauses that represent both a data base of known facts and a set of rules to derive new facts out of known ones. For example one could use a relation named `eq-db` to link a type to its equality test.

```
eq-db "nat" "nat_eq".
eq-db (app["list", B]) (app["list_eq", B, B_eq]) :-
  eq-db B B_eq.
```

The first clause is a fact stating that `nat_eq` is the equality test for type `nat`. The second clause is an inference one and reads: the equality test for `(list B)` is `(list_eq B B_eq)` if `B_eq` is the equality test for `B`.

The `eq-db` data base can be queried for an equality test for, say, `(list nat)` as follows:

```
eq-db (app["list", "nat"]) F.
```

where `F` is a variable to be filled in. By chaining the two clauses Elpi answers:

```
F = app["list_eq", "nat", "nat_eq"]
```

that read back in the Coq syntax is `(list_eq nat nat_eq)`, the desired equality test.


It is worth recalling out that in λ Prolog the set of clauses is dynamic: a program is allowed to add clauses inside a specific scope (typically the one of a binder) and the runtime collects them when the scope ends. As we will see, this feature is useful when a derivation takes place under an hypothetical context, e.g. when one assumes a parameter `A` and an equality test `A_eq`. No other feature of the Elpi language is relevant to this paper.

²In this paper we simplify a little the embedding and use strings to represent Coq constants. In reality global constants are explicit nodes, e.g. `nat`, being an inductive type, is written `(indt "Coq.Init.Datatypes.nat")`, while `fact`, being a constant, is written `(const "Coq.Arith.Factorial.fact")`.

Finally, the integration of Elpi in Coq exposes to the extension language primitives to access the logical environment, e.g. to read an inductive data type declaration; to declare a new inductive type; to define a new constant; etc...

5 Anatomy of the derivation

The structure of the derivation is depicted in the following diagram. Each box represents a component deriving a complete term. An arrow from component A to component B tells that the terms generated by B are used by the terms generated by A. The interfaces between these components are indeed types: one can replace the work done by each component with a few hand written terms, if necessary.



The `eq` component is in charge of synthesizing the program performing the equality test.

The correctness proof generated by `eqcorrect` goes by induction on the first term of the two being compared and then goes on in a different branch for each constructor `K`. The property being proved by induction is expressed using `eq_axiom` that, as we will detail in section 5.6 is equivalent to a double implication. The `bcongr` component proves that the property is preserved by equal contexts, that is when the two terms are built using the same constructor. When they are not the program must return false and the equality be false as well: this is shown by `eqK`, that performs the case split on the second term. The no confusion property of constructor is key to this contextual reasoning. `projK` and `isK` generate utility functions that are then used by `injection` and `discriminate` to prove that constructors are injective and different. As we sketched in the previous sections the unary parametricity translation plays a key role in expressing the induction principle. The inductive predicate `is_T` for an inductive type `T` is generated by `param1` while `param1P` shows that terms of type `T` validate `is_T`. `map` shows that `is_T` is a functor when `T` has parameters. This property is both used to synthesize induction principles and also to combine the pieces together in the correctness proof. The `eqOK` component hides the `is_T` relation from the theorems proved by `eqcorrect` by using the lemmas `T_is_T` proved by `param1P`.

5.1 Equality test

Synthesizing the equality test for a type τ is the simplest step. For each type parameter A an equality test A_eq has to be taken in input. Then the recursive function takes in input two terms of type τ and inspects both via pattern matching. Outside the diagonal, where constructors are different, we return `false`. On the diagonal we compose the calls on the arguments of the constructors using boolean conjunction. The code called to compare two arguments depends on their type. If it is τ then it is a recursive call. If it is a type parameter A then we use A_eq . If it is another type constructor we use the equality test for it.

Lets take for example the equality test for rose trees:

```
1 Definition rtree_eq A (A_eq : A → A → bool) :=
2   fix rec (t1 t2 : rtree A) {struct t1} : bool :=
3     match t1, t2 with
4     | Leaf a, Leaf b => A_eq a b
5     | Node l, Node s => list_eq (rtree A) rec l s
6     | _, _ => false
7   end.
```

Line 5 calls `list_eq` since the type of `l` and `s` is `(list (rtree A))` and it passes to it `rec` since the type parameter of `list` is `(rtree A)`.

Here is an excerpt of Elpi code used to synthesise the body of the branches:

```
eq-db "A" "A_eq".
eq-db (app["rtree","A"]) "rec".
eq-db (app["list", B]) (app["list_eq", B, B_eq]) :-
  eq-db B B_eq.
```

The first clause says that A_eq is the equality test for type A , and is used to build the branch at line 4. The third clause, chained with the second one, combines `list_eq` with `rec` building the branch at line 5.

The first two clauses are present only during the derivation of the fixpoint, under the context formed by the type parameter A and its equality test A_eq . Once the derivation is complete both clauses are removed from the data base and the following one is permanently added.

```
eq-db (app["rtree", B]) (app["rtree_eq", B, B_eq]) :-
  eq-db B_eq.
```

5.2 Parametricity

The `[pram1]` component is able to generate the unary parametricity translation of types and terms following [4]. We already gave a few examples in section 3, we repeat here just the one for rose trees:

```
Inductive is_rtree A (PA : A → U) : rtree A → U :=
| is_Leaf a (pa : PA a) : is_rtree A PA (Leaf A a)
| is_Node l (pl : is_list (rtree A) (is_rtree A PA) l) :
  is_rtree A PA (Node A l).
```

The `[pram1P]` component synthesizes proofs that terms of type τ validate is_T by a trivial structural recursion: constructor K is mapped to is_K .

```
Definition rtree_is_rtree A (PA : A → U) :
  (∀ x, PA x) → ∀ t, is_rtree A PA t.
```

5.3 Functoriality

The `[map]` component implements a double service.

For simple containers it synthesizes what one expects. For example:

```
Definition rtree_map A1 A2 :
  (A1 → A2) → rtree A1 → rtree A2.
```

The derivation on containers with no indexes is useful in general but is not needed in order to synthesize equality tests nor their correctness proofs. On the contrary it becomes crucial when the container has indexes, e.g. when the container is a `is_T` inductive predicate.

On indexed data types the derivation avoids to map the indexes and consequently all type variables occurring in the types of the indexes. For example, mapping the `is_list` inductive predicate gives:

```
Lemma is_list_map : A PA PB,
  (∀ a, PA a → PB a) →
  ∀ l, is_list A PA l → is_list A PB l.
```

This property corresponds to the functoriality of `is_list` over the property about the type parameter. Note that parameters of arity one, such as PA , are mapped point wise.

As we did for the `eq-db` data base of equality tests, we can store these maps as clauses and use the data base later on in the `[induction]` and `[eqcorrect]` derivations. Here is an excerpt of Elpi code for this data base, that we call `map-db`:

```
map-db (app["is_list", A, PA])
  (app["is_list", A, PB])
  (app["is_list_map", A, PA, PB, F]) :-
  map-db PA PB F.
```

Note that the terms involved are “point free”, i.e. the first two arguments are terms of arity one, while the third term is of arity two. For example the identity map would be written as follows:

```
map-db PA PA (lam (λ a, lam (λ pa, pa))).
```

This means that when one has a term a and a term $(pa : PA a)$, in order to obtain a term $(qa : QA a)$ he can query `map-db` as follows:

```
map-db "PA" "QA" M
```

If the answer is $M = f$ then the desired term is obtained by passing a and pa to f , i.e. $(f a pa : QA a)$.

5.4 Induction

In order to derive the induction principle for type τ we first derive its unary parametricity translation is_T .

The `is_T` inductive predicate has one constructor `is_K` for each constructor K of the type τ . The type of `is_K` relates to the type of K in the following way. For each argument $(a : A)$ of K , `is_K` takes two arguments: $(a : A)$ and $(pa : is_A a)$. Finally the type of $(is_K a1 pa1 .. an pan)$ is $(is_T (K a1 .. an))$.

The induction principle can be synthesized as follows:

1. take in input each parameter $A1\ PA1 \dots An\ PAn$ of is_T .
2. take in input a predicate $(P : T\ A1 \dots An \rightarrow U)$.
3. for each constructor is_K of type $(\forall A1\ PA1 \dots An\ PAn, \forall a1\ pa1 \dots am\ pam, is_T\ A1\ PA1 \dots An\ PAn\ (K\ a1 \dots am))$ take in input an assumption HK of type $(\forall a1\ pa1 \dots am\ pam, P\ (K\ a1 \dots am))$.
4. take in input $(t : T\ A1 \dots An)$.
5. take in input $(x : is_T\ A1\ PA1 \dots An\ PAn)$.
6. perform recursion on x and a case split. Then in each branch
 - a. bind all arguments of is_K , namely $(a1 : A1)$
 $(pa1 : is_A1\ a1) \dots (an : An) (pan : is_An\ an)$
 - b. obtain qai by *mapping* the corresponding pai (as in `map-db`, see below).
 - c. return $(HK\ a1\ qai \dots an\ qan)$

Lets take for example the induction principle for rose trees:

```

Definition rtree_induction A PA P
  (HLeaf :  $\forall a, PA\ a \rightarrow P\ (Leaf\ A\ a)$ )
  (HNode :  $\forall l, is\_list\ (rtree\ A)\ P\ l \rightarrow P\ (Node\ A\ l)$ ) :
 $\forall t, is\_rtree\ A\ PA\ t \rightarrow P\ t$ 
:=
fix IH (t : rtree A) (x : is_rtree A PA t) {struct x}: P t :=
match x with
| is_Leaf a pa => HLeaf a pa
| is_Node l pl =>
  (* pl : is_list (rtree A) (is_rtree A PA) l *)
  HNode l
  (is_list_map (rtree A) (is_rtree A PA) P IH l pl)
end.
```

Note how, intuitively, the type of `HLeaf` can be obtained from the type of `is_Leaf` by replacing $(is_rtree\ A\ PA)$ with P .

Finally lets see how the second argument to `HNode` is synthesized. We take advantage of the fact that Elpi is a logic programming language and we query the data base `map-db` as follows. First we temporarily register the fact that `IH` maps $(is_rtree\ A\ PA)$ to P obtaining, among others, the following clauses.

```

map-db (app["is_rtree", "A", "PA"]) "P" "IH".
map-db (app["is_list", A, PA])
  (app["is_list", A, PB])
  (app["is_list_map", A, PA, PB, F]) :-
  map-db PA PB F.
```

Then we query `map-db` as follows:

```

map-db (app["is_list", app["rtree", "A"],
  app["is_rtree", "A", "PA"]])
  (app["is_list", app["rtree", "A"],
    "P"])
```

Q.

The answer

```

Q = app["is_list_map", app["rtree", "A"],
  app["is_rtree", "A", "PA"], "P", "IH"]
```

is exactly the second term we need to pass to `HNode` (once applied to `l` and `pl`).

It is worth pointing out that, for the term to be accepted by the termination checker the `map` over `is_list` must be transparent.

To sum up the unary parametricity translation gives us the type of the induction principle, up to a trivial substitution. The functoriality property of the inductive predicates obtained by parametricity gives us a way to prove the branches.

5.5 No confusion property

In order to prove that an equality test is correct one has to show the so called “no confusion” property, that is that constructors are injective and disjoint (see for example [7]).

Let’s start by proving they are disjoint. The simplest form of this property can be expressed on `bool`:

```

Lemma bool_discr : true = false  $\rightarrow \forall T : U, T$ .
```

This lemma is proved by hand once and for all. What the `isK` component synthesizes is a per-constructor test to be used in order to reduce a discrimination problem on type T to a discrimination problem on `bool`. For the rose tree data type `isK` generates the following constants:

```

Definition rtree_is_Node A (t : rtree A) : bool :=
  match t with Node _ => true | _ => false end.
```

```

Definition rtree_is_Leaf A (t : rtree A) : bool :=
  match t with Leaf _ => true | _ => false end.
```

The `discriminate` components uses one more trivial fact, `eq_f` in order to assemble these tests together with `bool_discr`.

```

Lemma eq_f T1 T2 (f : T1  $\rightarrow$  T2) :
 $\forall a\ b, a = b \rightarrow f\ a = f\ b$ .
```

From a term H of type $(Node\ l = Leaf\ a)$ the `discriminate` procedure synthesizes a term of type $(\forall T : U, T)$ as follows:

```

1 bool_discr
2 (eq_f (rtree A) (rtree A) (rtree_is_Node A) H)
```

Note that the type of the term on line 2 is:

```

rtree_is_Node A (Node l) = rtree_is_Node A (Leaf a)
```

that is convertible to $(true = false)$.

In order to prove the injectivity of constructors the `projK` component synthesizes a projector for each argument of each constructor. For example

```

Definition list_get_cons1 A (d1 : A) (d2 : list A)
  (l : list A) : A :=
  match l with nil => d1 | cons x _ => x end.
```

```

Definition list_get_cons2 A (d1 : A) (d2 : list A)
  (l : list A) : list A :=
  match l with nil => d2 | cons _ xs => xs end.
```

Each projector takes in input default values for each and every argument of the constructor. It is designed to be used by the `injection` procedure as follows. Given a term H of type $(cons\ x\ xs = cons\ y\ ys)$, in order to obtain a term of type $(xs = ys)$ it generates:

```
eq_f H (list_get_cons2 A x xs)
```

This term is easy to build given that the type of `H` contains the default values to be passed to the projector. Note that the type of the entire term is:

```
list_get_cons2 A x xs (cons x xs) =
list_get_cons2 A x xs (cons y ys)
```

that is convertible to the desired $(xs = ys)$.

5.6 Congruence and `reflect`

In the definition of `eq_axiom` we used the `reflect` predicate [6]. It is a form of if-and-only-if specialized to link a proposition and a boolean test. It is defined as follows:

```
Inductive reflect (P : U) : bool → U :=
| ReflectT (p : P) : reflect P true
| ReflectF (np : P → False) : reflect P false.
```

To prove the correctness of equality tests the shape of `P` is always an equation between two terms of the inductive type, i.e. constructors. When the equality test finds the same constructor on both sides, as in $(k\ x_1 \dots x_n = k\ y_1 \dots y_n)$, it calls the appropriate equality tests for the arguments and forgets about the constructor. The `bcongr` component synthesizes lemmas helping to prove the correctness of this step. For example:

```
Lemma list_bcongr_cons A :
  ∀ (x y : A) b, reflect (x = y) b →
  ∀ (xs ys : list A) c, reflect (xs = ys) c →
  reflect (x :: xs = y :: ys) (b && c)

Lemma rtree_bcongr_Leaf A (x y : A) b :
  reflect (x = y) b → reflect (Leaf A x = Leaf A y) b

Lemma rtree_bcongr_Node A (l1 l2 : list (rtree A)) b :
  reflect (l1 = l2) b → reflect (Node A l1 = Node A l2) b
```

Note that these lemmas are not related to the equality test specific to the inductive type. Indeed they deal with the `reflect` predicate, but not with the `eq_axiom` that we use every time we talk about equality tests.

The derivation goes as follows: if any of the premises is false, then the result is proved by `ReflectF` and the injectivity of constructors. If all premises are `ReflectT` their argument, an equation, can be used to rewrite the conclusion.

```
1 Lemma list_bcongr_cons A
2   (x y : A) b (hb : reflect (x = y) b)
3   (xs ys : list A) c (hc : reflect (xs = ys) c) :
4   reflect (x :: xs = y :: ys) (b && c) :=
5   match hb, hc with
6   | ReflectT eq_refl, ReflectT eq_refl => ReflectT eq_refl
7   | ReflectF (e : x = y → False), _ =>
8     ReflectF (λ H : (x :: xs) = (y :: ys) =>
9       e (eq_f (list A) A (list_get_cons1 A x xs)
10         (x :: xs) (y :: ys) H))
11   | _, ReflectF e => .. list_get_cons2 ..
12   end.
```

Note how the elimination of `reflect` substitutes the boolean expression by either `true` or `false`. Inside the branch at line 6 the boolean expression is hence $(true \ \&\& \ true)$ while the proposition is $(x :: xs = x :: xs)$ given that the two equations $(x = y)$ and $xs = ys$ were eliminated.

Remark that the argument of `e` at line 9 is the term generated by the `injection` component. The branch at line 11, covering the case where the heads are equal but the tails different, is very close to lines 9 and 10 but for the fact that the projector for the second argument of `cons` is used, instead of the first one.

There are other ways one could have expressed these lemmas, for example by not mentioning the `cons` constructor explicitly but rather an abstract function `k` known to be injective on the first and second argument. Even if we find this presentation more appealing on paper, in practice we found no advantage and we hence opted for the current approach where the statements mention the constructor directly.

5.7 Congruence and `eq_axiom`

The `bcongr` component gives us lemmas to propagate equality and inequality under the same constructor. The component we describe here, `eqK`, pattern matches on the second term and either appeals to the lemma generated by `bcongr` or proves `eq_axiom` with `ReflectF`.

Recall that the first term is already being analysed by the induction principle. `eqK` generates a lemma for each constructor, to be used in the corresponding branch of the induction. This is the one for `Node`:

```
Lemma rtree_eq_axiom_Node A (A_eq : A → A → bool) l1 :
  eq_axiom (list (rtree A))
    (list_eq (rtree A) (rtree_eq A A_eq)) l1 →
  eq_axiom (rtree A) (rtree_eq A A_eq) (Node A l1)
:=
  λ H (t2 : rtree A) =>
  match t2 with
  | Leaf n =>
    ReflectF (λ abs : Node A l1 = Leaf A n =>
      bool_discr
        (eq_f (rtree A) bool (rtree_is_Node A)
          (Node A l1) (Leaf A n) abs)
      False)
  | Node l2 =>
    rtree_bcongr_Node A l1 l2
    (list_eq (rtree A) (rtree_eq A A_eq) l1 l2) (H l2)
  end.
```

Note that the code for the first branch is what `discriminate` synthesizes; while the code in the second branch is what `bcongr` generates.

5.8 Correctness

The `eqcorrect` component combines the induction principle generated by `induction` with the case split on the second term provided by `eqK`.

Let's recall the type of the correctness lemma for `list_eq`, of the induction principle and then let's analyse the proof of `rtree_eq_correct`:

```

Lemma list_eq_correct A (fa : A → A → bool) l,
  is_list A (eq_axiom A fa) l →
  eq_axiom (list A) (list_eq A fa) l.

Definition rtree_induction A PA P
  (HLeaf : ∀y, PA y → P (Leaf A y))
  (HNode : ∀l, is_list (rtree A) P l → P (Node A l)) :
  ∀t, is_rtree A PA t → P t.

Lemma rtree_eq_axiom_Node A (f : A → A → bool) l1 :
  eq_axiom (list (rtree A))
    (list_eq (rtree A) (rtree_eq A f)) l1 →
  eq_axiom (rtree A) (rtree_eq A f) (Node A l1).

```

The proof is a rather straightforward application of the induction principle to the property

```
eq_axiom (rtree A) (rtree_eq A fa)
```

Each branch is then proved by the corresponding lemma generated by `eqK` with only one caveat: one may need to adapt the induction hypothesis, `P1` here, in order to make it fix the premise of the lemma generated by `eqK`. In this specific case the "adaptor" is `list_eq_correct`.

```

Lemma rtree_eq_correct A (fa : A → A → bool) :=
  rtree_induction A (eq_axiom A fa)
  (*P*) (eq_axiom (rtree A) (rtree_eq A fa))
  (*HLeaf*) (rtree_eq_axiom_Leaf A fa)
  (*HNode*) (λl (P1 : is_list (rtree a)
    (eq_axiom (rtree a) (rtree_eq a fa)) l) =>
    rtree_eq_axiom_Node A fa l
    (list_eq_correct (rtree a) (rtree_eq a fa) l P1)).

```

Logic programming provides again a natural way to synthesize the adaptor. In particular we use `map-db` to find the link as follows. We load in the data base all the correctness proofs synthesized so far, as follows:

```

map-db (app["is_list", A,
  PA])
  (app["eq_axiom", app["list", A],
    app["list_eq", A, A_eq]])
  (app["list_eq_correct", A, A_eq]) :-
  map-db PA (app["eq_axiom", A, A_eq]).

```

This clause simply gives an operational reading to the type of `list_eq_correct`: the conclusion is true if the premise is. The only cleverness is to separate the premise in two parts, being a `(list A)` with property `PA` and have `PA` be a sufficient condition to prove that `A_eq` is correct. In this way clauses compose better, e.g. the inference step peels off just one type constructor at a time.

We extend the `map-db` predicate, instead of building a new one just for correctness lemmas, because functoriality lemmas are sometimes needed in addition to the correctness ones. Take for example this simple data type of an histogram.

```
Inductive histogram := Columns (bars : list nat).
```

```

Lemma histogram_induction (P : histogram → Type) :
  (∀l, is_list nat is_nat l → P (Columns l)) →
  ∀h, is_histogram h → P h.

```

Now look at the lemma synthesized by `eqK` for the `Columns` constructor.

```

Lemma histogram_eq_axiom_Columns l :
  eq_axiom (list nat) (list_eq nat nat_eq) l →
  ∀h, eq_axiom_at histogram histogram_eq (Columns l) h.

Lemma histogram_eq_correct h :
  eq_axiom histogram histogram_eq h
:=
  histogram_induction
    (eq_axiom histogram histogram_eq)
    (λl (P1 : is_list nat is_nat l) =>
      histogram_eq_axiom_Columns
        l (list_eq_correct nat nat_eq
          l (is_list_map nat
            is_nat (eq_axiom nat nat_eq)
            nat_eq_correct l P1))).

```

Note that the type of `P1` is `(is_list nat is_nat)` and that it needs to be adapted to match `(is_list nat (eq_axiom nat nat_eq))`. The correctness lemma `nat_eq_correct` cannot be used directly but must undergo the `is_list` functor.

5.9 eqOK

The last derivation hides the `is_T` predicate to the final user by combining the output of `eqcorrect` and `param1P`.

```

Lemma list_eq_correct A A_eq :
  ∀l, is_list A (eq_axiom A A_eq) l →
  eq_axiom list A (list_eq A A_eq) l.

Lemma list_eq_OK A A_eq (HA : ∀a, eq_axiom A A_eq a) l :
  eq_axiom list A (list_eq A A_eq) l
:=
  list_eq_correct A A_eq l (list_is_list A HA).

```

Both lemmas need to be available: the former composes well and may be needed if one defines a type using rose trees as a container. The latter is what the user needs to work with rose trees.

5.10 Assessment

The code is quite compact thanks to the fact that the programming language is very high level and that its programming paradigm is a good fit for this application.

On the average each components is about 200 lines of code. Simpler derivations like `projK`, `isK` or even `param1P` are under 100 lines.

Debugging this kind of code did not pose particular difficulties. The typical error results in the generated term being ill-typed. In that case the Coq type checker could be used to identify the culprit. Given how small the bricks are, it was simple to identify the lines generating the offending sub-term.

The time required to design and develop the entire procedure amounts to approximatively six months, but spanned over more than one and a half year: most of the time has been spent improving the integration of Elpi in Coq in response to the experience gathered on this work.

5.10.1 Incompleteness and user intervention

At the time of writing the Elpi integration in Coq does not support mutual inductive types, universe polymorphic definitions and primitive projections.

All derivations support polynomial types. Some derivations also support index data, eg `eq` is able to synthesize an equality test for vectors. Most of the derivations for contextual reasoning, such as `eqK` and `bcongr` do not support indexes. Some do, for example `projK` derives this projector for the last component of the `cons` constructor a vector:

```
Definition vector_get_cons3 A n
  (d1 : A) (d2 : nat) (d3 : Vector.t A d2) :
  Vector.t A n → {m : nat & Vector.t A m}.
```

It is folklore that if the type of indexes, `nat` here, admits an equality test then the dependent pair can be unpacked without loosing information and that two such dependent pairs can be equated without major difficulties.

Given that the output of each component in the derivation can be replaced by user provided terms we tried to fill the gap by hand. It required around 20 lines of boilerplate to link the “wrapped” vectors to the regular ones and other 20 lines to perform what `bcongr` and `eqK` could, in principle, do. It hence looks doable to extend the derivation to cover this class of index data types in the future.

6 Related work

Systems similar to Coq [16], e.g. Matita [1], Lean [2], Agda [11] and Isabelle [10] all generate induction principles automatically, and some of them also the no confusion properties.

To our knowledge they do not generate sensible induction principles when containers are involved and do not generate proved equality tests out of the box.

Most of the systems cited above come with simple forms of Prolog-like automation, usually in the form of type classes. The user typically resorts to that in order to perform some of the inductive reasoning one needs in order to synthesise code in a type directed way. To our knowledge no ready-to-use package to synthesise equality tests and their proofs was written this way.

Some systems, notably Lean, come with a whole round meta programming framework. Still, to our knowledge, the primary application is the development of proof commands, not program/proof synthesis, in spite of the stunning similarity.

Coq provides two mechanisms strictly related to this work.

The `Scheme Equality` command generates for a type `τ` the code for the equality test (`τ_eqb`) and a proof that equality

is decidable on `τ`. The proof internally uses the equality test, but its type does not:

```
τ_eq_dec : ∀ x y : τ, {x = y} + {x <> y}
```

By unfolding the proof term, that is transparent, it should be possible to recover the fact that `τ_eqb` is a correct equality test. Data types defined using containers are not supported.

The `decide equality` tactic requires the user to start a lemma with a statement as the one depicted above. The tactic only performs one (case split) step and has to be iterated by hand. It does not remember which equalities were proved decidable before, it is up to the user to eventually share code. The proof term generated is, in a type theoretic sense, a program even if its code mixes the comparison test with its correctness proof. This proof is fully transparent, and inlines all the contextual reasoning steps such as injection and discrimination. As a result the term is very large and computationally heavy when run within Coq.

In the programming language world derivation is much more developed. The dominant approach is to provide some meta programming facilities, e.g. by providing a syntactic declaration of types and then use the programming language itself to write derivations [13] that run at compile time as compiler plugins.

Our approach is similar in a sense, since we work at the meta level on the syntax of types (and terms), but it is also very different since we pick a different programming language for meta programming. In particular we choose a very high level one that makes our derivations very concise and hides uninteresting details such as the representation of bound variables. The derivation described in the paper is the result of many failed attempts and we believe that the high level nature of programming language we chose played an important role in the exploratory phase.

7 Conclusion

We described a technique to define stronger induction principles for Coq data types built using containers. We use the unary parametricity translation in order to separate terms from types, express structural properties and finally confine the modularity problems stemming from the termination check implemented in Coq. Finally we provide a Coq package deriving correct equality tests for polynomial inductive data types.

It seems reasonable to extend the current derivation code to cover inductive types with decidable indexes, as hinted in section 5.10.1. For types not supported by the derivation it should be possible to improve the way user intervention is requested: right now errors are printed, but the exact type of the missing derivation has to be written down, and of course proved, by the user.

We also look forward to let the user tune the derivation process by annotating the type declarations. For example

the user may want to skip certain arguments when generating the equality test, such as the integer describing the length of a sub vector in the `cons` constructor. The resulting equality test surely requires some user intervention in order to be proved correct, but it features a better computational complexity.

Finally, adding other derivations seems appealing. For example the interface next to `eqType` in the hierarchy used in the Mathematical Component library is the one of countable types, i.e. types in bijection with natural numbers. The interface requires, roughly, a serialization function to another countable type, a tedious task that could be made automatic.

Acknowledgments

We are grateful to Maxime Denes and Cyril Cohen for the many discussions shedding light on the subject. We thank Cyril Cohen for writing the code of `[param2]` (binary parametricity translation), out of which `[param1]` was easily obtained. We also thank Damien Rouhling and Laurent Théry for proofreading the paper. Finally we are indebted to Luc Chabassier for working on an early prototype of Elpi on the synthesis of equality tests: an experiment that convinced the author it was actually doable.

References

- [1] Andrea Asperti, Wilmer Ricciotti, Claudio Sacerdoti Coen, and Enrico Tassi. 2011. The Matita Interactive Theorem Prover. In *Automated Deduction – CADE-23*, Nikolaj Bjørner and Viorica Sofronie-Stokkermans (Eds.). Springer Berlin Heidelberg, Berlin, Heidelberg, 64–69.
- [2] Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris van Doorn, and Jakob von Raumer. 2015. The Lean Theorem Prover (System Description). In *Automated Deduction – CADE-25*, Amy P. Felty and Aart Middeldorp (Eds.). Springer International Publishing, Cham, 378–388.
- [3] Cvetan Dunchev, Ferruccio Guidi, Claudio Sacerdoti Coen, and Enrico Tassi. 2015. ELPI: fast, Embeddable, λ Prolog Interpreter. In *Proceedings of LPAR*. Suva, Fiji. <https://hal.inria.fr/hal-01176856>
- [4] Chantal Keller and Marc Lasson. 2012. Parametricity in an Impredicative Sort. In *CSL - 26th International Workshop/21st Annual Conference of the EACSL - 2012 (CSL)*, Patrick Cégielski and Arnaud Durand (Eds.), Vol. 16. Schloss Dagstuhl - Leibniz-Zentrum fuer Informatik, Fontainebleau, France, 381–395. <https://doi.org/10.4230/LIPIcs.CSL.2012.399>
- [5] Assia Mahboubi and Enrico Tassi. 2013. Canonical Structures for the Working Coq User. In *Interactive Theorem Proving*, Sandrine Blazy, Christine Paulin-Mohring, and David Pichardie (Eds.). Springer Berlin Heidelberg, Berlin, Heidelberg, 19–34.
- [6] Assia Mahboubi and Enrico Tassi. 2018. *Mathematical Components*. draft, v1-183-gb37ad7.
- [7] Conor McBride, Healfdene Goguen, and James McKinna. 2006. A Few Constructions on Constructors. In *Types for Proofs and Programs*, Jean-Christophe Filliâtre, Christine Paulin-Mohring, and Benjamin Werner (Eds.). Springer Berlin Heidelberg, Berlin, Heidelberg, 186–200.
- [8] Dale Miller. 2000. Abstract Syntax for Variable Binders: An Overview. In *Computational Logic – CL 2000*, John Lloyd, Veronica Dahl, Ulrich Furbach, Manfred Kerber, Kung-Kiu Lau, Catuscia Palamidessi, Luís Moniz Pereira, Yehoshua Sagiv, and Peter J. Stuckey (Eds.). Springer Berlin Heidelberg, Berlin, Heidelberg, 239–253.
- [9] Dale Miller and Gopalan Nadathur. 2012. *Programming with Higher-Order Logic*. Cambridge University Press. <https://doi.org/10.1017/CBO9781139021326>
- [10] Tobias Nipkow, Markus Wenzel, and Lawrence C. Paulson. 2002. *Isabelle/HOL: A Proof Assistant for Higher-order Logic*. Springer-Verlag, Berlin, Heidelberg.
- [11] Ulf Norell. 2007. *Towards a practical programming language based on dependent type theory*. Ph.D. Dissertation. Department of Computer Science and Engineering, Chalmers University of Technology, SE-412 96 Göteborg, Sweden.
- [12] Jorge Luis Sacchini. 2011. *On type-based termination and dependent pattern matching in the calculus of inductive constructions*. Theses. École Nationale Supérieure des Mines de Paris. <https://pastel.archives-ouvertes.fr/pastel-00622429>
- [13] Tim Sheard and Simon Peyton Jones. 2002. Template Metaprogramming for Haskell. *SIGPLAN Not.* 37, 12 (Dec. 2002), 60–75. <https://doi.org/10.1145/636517.636528>
- [14] Matthieu Sozeau and Nicolas Oury. 2008. First-Class Type Classes. In *Proceedings of the 21st International Conference on Theorem Proving in Higher Order Logics (TPHOLs '08)*. Springer-Verlag, Berlin, Heidelberg, 278–293. https://doi.org/10.1007/978-3-540-71067-7_23
- [15] Enrico Tassi. 2018. Elpi: an extension language for Coq (Metaprogramming Coq in the Elpi λ Prolog dialect). (Jan. 2018). <https://hal.inria.fr/hal-01637063> CoqPL.
- [16] The Coq Development Team. 2018. The Coq Proof Assistant, version 8.8.0. <https://doi.org/10.5281/zenodo.1219885>
- [17] Philip Wadler. 1989. Theorems for Free!. In *Proceedings of the Fourth International Conference on Functional Programming Languages and Computer Architecture (FPCA '89)*. ACM, New York, NY, USA, 347–359. <https://doi.org/10.1145/99370.99404>